In[84]:=
```
ClearAll["Global`*"];
```

In[85]:=
```
y1 := L1 Cos[ϕ1[t]] ;
x1 := L1 Sin[ϕ1[t]] ;
y2 := y1 + L2 Cos[ϕ2[t]] ;
x2 := x1 + L2 Sin[ϕ2[t]];
x1dot := D[x1, t];
y1dot := D[y1, t];
x2dot := D[x2, t];
y2dot := D[y2, t];
T := (m1 / 2) (x1dot^2 + y1dot^2) + (m2 / 2) (x2dot^2 + y2dot^2)
FullSimplify[T]
```

Out[94]=
$$\frac{1}{2}\left(L1^2\ (m1+m2)\ \phi1'[t]^2+2\ L1\ L2\ m2\ \text{Cos}[\phi1[t]-\phi2[t]]\ \phi1'[t]\ \phi2'[t]+L2^2\ m2\ \phi2'[t]^2\right)$$

In[95]:=
```
(* note that I defined y axis to point downward *)
U := -m1 g y1 - m2 g y2
L := T - U
FullSimplify[L]
```

Out[97]=
$$\frac{1}{2}\left(2\ g\ (L1\ (m1+m2)\ \text{Cos}[\phi1[t]]+L2\ m2\ \text{Cos}[\phi2[t]])+\right.$$
$$\left.L1^2\ (m1+m2)\ \phi1'[t]^2+2\ L1\ L2\ m2\ \text{Cos}[\phi1[t]-\phi2[t]]\ \phi1'[t]\ \phi2'[t]+L2^2\ m2\ \phi2'[t]^2\right)$$

In[98]:=
```
eomexpr1 := D[D[L, ϕ1'[t]], t] == D[L, ϕ1[t]]
eom1 := FullSimplify[eomexpr1]
eom1
```

Out[100]=
$$L1\left(L2\ m2\ \text{Sin}[\phi1[t]-\phi2[t]]\ \phi2'[t]^2+\right.$$
$$\left.(m1+m2)\ (g\ \text{Sin}[\phi1[t]]+L1\ \phi1''[t])+L2\ m2\ \text{Cos}[\phi1[t]-\phi2[t]]\ \phi2''[t]\right)==0$$

In[101]:=
```
eomexpr2 := D[D[L, ϕ2'[t]], t] == D[L, ϕ2[t]]
eom2 := FullSimplify[eomexpr2]
eom2
```

Out[103]=
$$L2\ m2\ \left(g\ \text{Sin}[\phi2[t]]-L1\ \text{Sin}[\phi1[t]-\phi2[t]]\ \phi1'[t]^2+\right.$$
$$\left.L1\ \text{Cos}[\phi1[t]-\phi2[t]]\ \phi1''[t]+L2\ \phi2''[t]\right)==0$$

In[104]:=
```
L1 = 1 ; L2 = 1 ; m1 = 1 ; m2 = 1; g = 9.8;
a = 0.1; b = -Sqrt[2]; tmax = 10;
soln = NDSolve[
    {eom1, eom2,
     ϕ1'[0] == 0, ϕ2'[0] == 0,
     ϕ1[0] == a, ϕ2[0] == a b},
    {ϕ1[t], ϕ2[t]},
    {t, 0, tmax}];
phi1[t_] := Evaluate[ϕ1[t] /. First[soln]]
phi2[t_] := Evaluate[ϕ2[t] /. First[soln]]
Plot[{phi1[t], phi2[t]}, {t, 0, tmax}]
```

Out[109]=



In[110]:=
```
xx1[t_] := L1 Sin[phi1[t]];
yy1[t_] := -L1 Cos[phi1[t]];
xx2[t_] := xx1[t] + L1 Sin[phi2[t]];
yy2[t_] := yy1[t] - L2 Cos[phi2[t]];
bobradius = 0.03;
Manipulate[
 L1 = 1 ; L2 = 1 ; m1 = 1 ; m2 = 1; g = 9.8;
 soln = NDSolve[
   {eom1, eom2,
    ϕ1'[0] == 0, ϕ2'[0] == 0,
    ϕ1[0] == a, ϕ2[0] == a b},
   {ϕ1[t], ϕ2[t]},
   {t, 0, tmax}];
 phi1[t_] := Evaluate[ϕ1[t] /. First[soln]];
 phi2[t_] := Evaluate[ϕ2[t] /. First[soln]];
 Animate[
  Graphics[
   {
    Line[{{0, 0}, {xx1[t], yy1[t]}}],
    Disk[{xx1[t], yy1[t]}, bobradius],
    Line[{{xx1[t], yy1[t]}, {xx2[t], yy2[t]}}],
    Disk[{xx2[t], yy2[t]}, bobradius]
   },
   Axes → True, PlotRange → {{-2, 2}, {2, -2}}
  ], {t, 0, tmax}],
 {{a, 0.1}, -1.5, 1.5},
 {{b, Sqrt[2]}, -Sqrt[2], +Sqrt[2]},
 {{tmax, 10}, 0, 100}]
```

Out[115]=



In[116]:=
```
Manipulate[
 L1 = 1; L2 = 1; m1 = 1; m2 = 1; g = 9.8;
 soln = NDSolve[
   {eom1, eom2,
    ϕ1'[0] == 0, ϕ2'[0] == 0,
    ϕ1[0] == a, ϕ2[0] == a b},
   {ϕ1[t], ϕ2[t]},
   {t, 0, tmax}];
 phi1[t_] := Evaluate[ϕ1[t] /. First[soln]];
 phi2[t_] := Evaluate[ϕ2[t] /. First[soln]];
 Plot[{phi1[t], phi2[t]}, {t, 0, tmax}],
 {{a, 0.1}, -1.5, 1.5},
 {{b, Sqrt[2]}, -Sqrt[2], +Sqrt[2]},
 {{tmax, 10}, 0, 100}]
```

Out[116]=




In[80]:=
```
Clear[m, a, b, c, d, e, f, g, h, i];
m = {{a, b, c}, {d, e, f}, {g, h, i}};
MatrixForm[m]
Det[m]
```

Out[82]//MatrixForm=

$$\begin{pmatrix} a & b & c \\ d & e & f \\ g & h & i \end{pmatrix}$$

Out[83]= $-c\,e\,g + b\,f\,g + c\,d\,h - a\,f\,h - b\,d\,i + a\,e\,i$